* OPEN STATA OUTPUT FILE LOG *

log using "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\Krause & Park.Authority DIfferentials.MANUSCRIPT & DESCRIPTIVE RESULTS.08-07-2024.smcl"  

  
   

  

**** 2nd ITERATION OF "STATUS GROUP DIFFERENTIALS AS REFERENCE POINTS FOR FOSTERING DIVERSITY AND INCLUSION WITHIN THE U.S. FEDERAL CIVILIAN WORKFORCE" [KRAUSE & PARK, NOVEMBER 2023] ****






*** ACCESS DATABASE FOR THE PROJECT: FEVS DATA FROM 2010-2019 AND 'MATCHED' OPM DATA ****


use "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\2010-2019_DATA FINAL.11-19-2023.dta", replace 

   
   
**************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************
 
 

**** 0. COMPUTE OVERALL IN-GROUP/OUT-GROUP AGENCY EMPLOYMENT RATIO VARIABLES & LN TRANSFORMED VARIABLES *****


gen ratio_fem_tot_men_tot = female_count / male_count
*
gen ratio_min_tot_nmin_tot = minority_count / nonminority_count



** Add a "2.11" positive constant to ensure negative values of "diversity2" can be log computed [diversity2_min = -2.102011] 

gen lndiversity2zeroadj = ln(diversity2 + 2.11)
*
*
*** NOTE FOR APPENDIX E: ORGANIZATIONAL JUSTICE LATENT DEPENDENT VARIABLE ***

** Add a "2.47" positive constant to ensure negative values of "justice2" can be log computed [justice2_min = -2.46851] 

gen lnjustice2zeroadj = ln(justice2 + 2.47)




* BETWEEN-IDENTITY GROUP DIFFERENTIAL: AGGREGATE [ALL AGENCY EMPLOYEES] PASSIVE REPRESENTATION COVARIATES: [WOMEN SUPERVISORS / MEN SUPERVIORS] &  [MINORITY SUPERVISORS / NON-MINORITY SUPERVISORS]
gen ln_ratio_fem_tot_men_tot  = ln(ratio_fem_tot_men_tot)
*
gen ln_ratio_min_tot_nmin_tot = ln(ratio_min_tot_nmin_tot)
*
*
*
* BETWEEN-IDENTITY GROUP DIFFERENTIAL: FIXED STATUS GROUP [SUPERVISORS] PASSIVE REPRESENTATION COVARIATES: [WOMEN SUPERVISORS / MEN SUPERVIORS] &  [MINORITY SUPERVISORS / NON-MINORITY SUPERVISORS]
gen ln_ratio_fsup_msup         = ln(ratio_fsup_msup)
*
gen ln_ratio_minsup_nonmsup    = ln(ratio_minsup_nonmsup)
*
*
*
* WITHIN-'OUT-GROUP' STATUS DIFFERENTIAL PASSIVE REPRESENTATION COVARIATES: [WOMEN SUPERVISORS / WOMEN NON-SUPERVISORS] &  [MINORITY SUPERVISORS / MINORITY NON-SUPERVISORS]
gen ln_ratio_fsup_fsub      = ln(ratio_fsup_fsub)
*
gen ln_ratio_minsup_minsub  = ln(ratio_minsup_minsub)
*
*
*
* BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL PASSIVE REPRESENTATION COVARIATES: [WOMEN SUPERVISORS / MEN SUPERVISORS] / [WOMEN NON-SUPERVISORS / MEN NON-SUPERVISORS] & [MINORITY SUPERVISORS / NON-MINORITY SUPERVISORS] / [MINORITY NON-SUPERVISORS / NON-MINORITY NON-SUPERVISORS]
gen ln_ratio_fmsup_fmsub     = ln(ratio_fmsup_fmsub)
*
gen ln_ratio_mnmsup_mnmsub   = ln(ratio_mnmsup_mnmsub)
*
*
*
* GENERATE LOGGED VERSION OF AGENCY PROFESSIONALISM CONTROL COVARIATE *
gen ln_professionals_total_ratio  = ln(professionals_total_ratio)
 
*
*
*

* GENERATE MINORITY WOMEN RESPONSE BINARY INDICATOR (I.E., GENDER ==1 & MINORITY==1)
gen minority_women = 1 if minority==1 & gender==1
replace minority_women = 0 if minority_women ==. 
*
*

* GENERATE WITHIN SOCIAL-IDENTITY GROUP RESPONDENT HETEROGENEITY TRICHOTMOUS INDICATOR (I.E., MINORITY WOMEN RESPONDENT = 2; WHITE WOMEN/MINORITY MEN RESPONDENT = 1; BASELINE CATEGORY: MEN/NON-MINORITY RESPONDENTS = 0)
gen     women_het = 2 if gender==1 & minority==1 
replace women_het = 1 if gender==1 & minority==0 
replace women_het = 0 if women_het==.
*
*
gen     minority_het = 2 if minority==1 & gender==1 
replace minority_het = 1 if minority==1 & gender==0 
replace minority_het = 0 if minority_het==. 
 
 

***************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************
 

 
 

**** 1. PRELIMINARY DATA FEATURES *******



*** TABLE A1: DESCRIPTIVE STATISTIC: BASED ON EFFECTIVE REGRESSION SAMPLES ***

quietly regress  lndiversity2zeroadj     gender minority  supervisor  topoffgender_2 lntotworkforce_count  ln_professionals_total_ratio   i.agencyid i.year, vce(cluster agencyid)

sum diversity2 lndiversity2zeroadj   justice2  lnjustice2zeroadj         ratio_min_tot_nmin_tot ln_ratio_min_tot_nmin_tot    ratio_fsup_msup  ln_ratio_fsup_msup  ratio_minsup_nonmsup  ln_ratio_minsup_nonmsup ///
ratio_fsup_fsub  ln_ratio_fsup_fsub   ratio_minsup_minsub  ln_ratio_minsup_minsub   ratio_fmsup_fmsub   ln_ratio_fmsup_fmsub   ratio_mnmsup_mnmsub  ln_ratio_mnmsup_mnmsub  gender minority supervisor  lntotworkforce_count  professionals_total_ratio ln_professionals_total_ratio    topoffgender_2 topoffminority_2 if e(sample), detail
*
tab gender  if e(sample)
tab minority if e(sample)
*
tab gender minority if e(sample)
tab minority_women if e(sample)
*
tab women_het if e(sample)
tab minority_het if e(sample)
*
*
*
tab supervisor if e(sample)
*
*
tab gender supervisor if e(sample)
tab minority supervisor if e(sample)
*
*



   
*** MANUSCRIPT NOTES: SIMPLE BIVRIATE CORRELATIONS AMONG OVERALL PR, SUPERVISORY PR, ABSOLUTE SGPD, & RELATIVE SGPD MEASURES (BOTH LEVELS AND NATURAL LOGARITHM FORMATS) ***   


*** EVALUATE SIMILARITIES/DISSIMILARITIES AMONG ALTERNATIVE MEASURES OF DESCRIPTIVE REPRESENTATION WITHIN U.S. FEDERAL AGENCIES ***


*** VARIABLES OPERATIONALIZED IN LEVELS ***

correlate ratio_fem_tot_men_tot    ratio_fsup_msup   ratio_fsup_fsub   ratio_fmsup_fmsub 
*
correlate ratio_min_tot_nmin_tot  ratio_minsup_nonmsup   ratio_minsup_minsub   ratio_mnmsup_mnmsub



*** VARIABLES OPERATIONALIZED IN NATURAL LOGARITHMS [REDUCE SKEW AND KURTOSIS] ***

correlate  ln_ratio_fem_tot_men_tot   ln_ratio_fsup_msup         ln_ratio_fsup_fsub      ln_ratio_fmsup_fmsub
*
correlate  ln_ratio_min_tot_nmin_tot  ln_ratio_minsup_nonmsup    ln_ratio_minsup_minsub   ln_ratio_mnmsup_mnmsub



*** TAKEAWAYS: (1) OVERALL PR AND SUPERVISORY PR MEASURES ARE STRONGLY CORRLATED AND THUS PICKING UP THE SAME DESCRIPTIVE REPRESENTATION CONCEPT WITH THESE SAMPLE OF DATA ***
***     (2) ABSOLUTE AND RELATIVE SGPD MEASURES ARE MODERATELY CORRELATED -- BUT SUFFICIENTLY DISTINCT MEASURES -- MAKES SENSE SINCE FORMER MEASURE IS "NUMERATOR" OF LATTER MEASURE
***     (3) ABSOLUTE AND RELATIVE SGPD MEASURES ARE OFTEN WEAKLY CORRELATED WITH OVERALL & SUPERVISORY PR MEASURES -- MEANS THAT SGPD MEASURES ARE CLEARLY DISTINCT FROM PR MEASURES    
 
 
 
********************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************



   
*** 2. TABLE A1: CONDITIONAL-RESPONDENT MODELS EVALUATING THE RELATIONSHIP INVOLVING WITHIN-IDENTITY "OUT-GROUP" STATUS & BETWEEN-IDENTITY GROUP STATUS DIFFERENTIALS AS A MEANS TO FOSTER DIVERSITY AND INCLUSION IN THE U.S. CIVILIAN WORKFORCE ***   



*********************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************


   
*** MODEL 1: CONDITIONAL RESPONSES BY GENDER -- GENDER BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL MODEL: [WOMEN SUPERVISORS WITHIN AGENCY j IN YEAR t / MEN SUPERVISORS WITHIN AGENCY j IN YEAR t] / [WOMEN NON-SUPERVISORS WITHIN AGENCY j IN YEAR t / MEN NON-SUPERVISORS  WITHIN AGENCY j IN YEAR t]  -- CONTROLLING FOR GENDER SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***


regress lndiversity2zeroadj  c.ln_ratio_fmsup_fmsub##i.gender  ln_ratio_fem_tot_men_tot  minority  supervisor  topoffgender_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year, vce(cluster agencyid)
*
estat ic
*

** BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN GENDERED RESPONDENTS **

lincom c.ln_ratio_fmsup_fmsub
*
*
lincom  1.gender#c.ln_ratio_fmsup_fmsub
*
*
*
*

   
*** MODEL 2: CONDITIONAL RESPONSES BY RACE/ETHNICITY -- RACIAL/ETHNIC BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL MODEL: [MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t] / [MINORITY NON-SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY NON-SUPERVISORS  WITHIN AGENCY j IN YEAR t] -- CONTROLLING FOR RACIAL/ETHNIC SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***


regress  lndiversity2zeroadj  c.ln_ratio_mnmsup_mnmsub##i.minority   ln_ratio_min_tot_nmin_tot    gender supervisor  topoffminority_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year, vce(cluster agencyid)
*
estat ic
*

** BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN MINORITY/NON-MINORITY RESPONDENTS **

lincom c.ln_ratio_mnmsup_mnmsub
*
*
lincom  1.minority#c.ln_ratio_mnmsup_mnmsub
*
*
*
*
*
*

  
*** MODEL 3: CONDITIONAL RESPONSES BY MINORITY WOMEN VERSUS WHITE WOMEN [BASELINE CATEGORY: MEN RESPONDENTS] -- GENDER BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL MODEL: [WOMEN SUPERVISORS WITHIN AGENCY j IN YEAR t / MEN SUPERVISORS WITHIN AGENCY j IN YEAR t] / [WOMEN NON-SUPERVISORS WITHIN AGENCY j IN YEAR t / MEN NON-SUPERVISORS WITHIN AGENCY j IN YEAR t]  -- CONTROLLING FOR GENDER SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***


regress lndiversity2zeroadj  c.ln_ratio_fmsup_fmsub##i.women_het    ln_ratio_fem_tot_men_tot   minority  supervisor  topoffgender_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year, vce(cluster agencyid)
*
estat ic
*

** BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN GENDERED RESPONDENTS **

lincom c.ln_ratio_fmsup_fmsub
*
*
lincom  1.women_het#c.ln_ratio_fmsup_fmsub
*
*
lincom  2.women_het#c.ln_ratio_fmsup_fmsub
*
*
lincom  2.women_het#c.ln_ratio_fmsup_fmsub -  1.women_het#c.ln_ratio_fmsup_fmsub
*
*
*
*


   
*** MODEL 4: CONDITIONAL RESPONSES BY MINORITY WOMEN VERSUS MINORITY MEN [BASELINE CATEGORY: NON-MINORITY RESPONDENTS] -- RACIAL/ETHNIC BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL MODEL: [MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t] / [MINORITY NON-SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY NON-SUPERVISORS WITHIN AGENCY j IN YEAR t] -- CONTROLLING FOR RACIAL/ETHNIC SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***

regress  lndiversity2zeroadj  c.ln_ratio_mnmsup_mnmsub##i.minority_het   ln_ratio_min_tot_nmin_tot    gender supervisor  topoffminority_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year if e(sample), vce(cluster agencyid)
*
estat ic
*
** BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN MINORITY/NON-MINORITY RESPONDENTS **

lincom c.ln_ratio_mnmsup_mnmsub
*
*
lincom  1.minority_het#c.ln_ratio_mnmsup_mnmsub
*
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub
*
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub -  1.minority_het#c.ln_ratio_mnmsup_mnmsub
*
*
*
*



************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************





   
*** 4. TABLE A1 [MODELS 5-8]: CONDITIONAL-RESPONDENT MODELS EVALUATING THE RELATIONSHIP INVOLVING WITHIN-IDENTITY "OUT-GROUP" STATUS & BETWEEN-IDENTITY GROUP STATUS DIFFERENTIALS AS A MEANS TO FOSTER DIVERSITY AND INCLUSION IN THE U.S. CIVILIAN WORKFORCE [BY NON-SUPERVISORS POSITIONS VERSUS SUPERVISORY POSITION] ***   





   
*** MODEL 5: CONDITIONAL RESPONSES BY GENDER & POSITION --  GENDER WITHIN-IDENTITY 'OUT-GROUP' STATUS DIFFERENTIAL MODEL: [WOMEN SUPERVISORS WITHIN AGENCY j IN YEAR t / WOMEN NON-SUPERVISORS WITHIN AGENCY j IN YEAR t] -- CONTROLLING FOR GENDER SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***

regress  lndiversity2zeroadj  c.ln_ratio_fmsup_fmsub##i.gender##i.supervisor    ln_ratio_fem_tot_men_tot   minority  topoffgender_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year, vce(cluster agencyid)
*
estat ic
*
** BY NON-SUPERVISORS RESPONDENT: WITHIN-IDENTITY "OUT" GROUP STATUS DIFFERENTIAL BETWEEN GENDERED RESPONDENTS **

lincom c.ln_ratio_fmsup_fmsub 
*
*
lincom  1.gender#c.ln_ratio_fmsup_fmsub

*
*
*
*

** BY SUPERVISOR RESPONDENT: WITHIN-IDENTITY "OUT" GROUP STATUS DIFFERENTIAL BETWEEN GENDERED RESPONDENTS **

lincom c.ln_ratio_fmsup_fmsub + 1.supervisor#c.ln_ratio_fmsup_fmsub
*
*
lincom  1.gender#c.ln_ratio_fmsup_fmsub +  1.gender#1.supervisor#c.ln_ratio_fmsup_fmsub






*** MODEL 6: CONDITIONAL RESPONSES BY RACE/ETHNICITIY & POSITION -- RACIAL/ETHNIC WITHIN-'OUT-GROUP' STATUS DIFFERENTIAL MODEL: [MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY NON-SUPERVISORS WITHIN AGENCY j IN YEAR t]  -- CONTROLLING FOR RACIAL/ETHNIC SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***

regress lndiversity2zeroadj  c.ln_ratio_mnmsup_mnmsub##i.minority##i.supervisor  ln_ratio_min_tot_nmin_tot   gender  topoffminority_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year, vce(cluster agencyid)
*
estat ic
*

** BY NON-SUPERVISORS RESPONDENT: WITHIN-IDENTITY "OUT" GROUP STATUS DIFFERENTIAL BETWEEN MINORITY/NON-MINORITY RESPONDENTS **

lincom c.ln_ratio_mnmsup_mnmsub
*
*
lincom  1.minority#c.ln_ratio_mnmsup_mnmsub

*
*
*
*

** BY SUPERVISOR RESPONDENT: WITHIN-IDENTITY "OUT" GROUP STATUS DIFFERENTIAL BETWEEN MINORITY/NON-MINORITY RESPONDENTS **

lincom c.ln_ratio_mnmsup_mnmsub + 1.supervisor#c.ln_ratio_mnmsup_mnmsub
*
*
lincom  1.minority#c.ln_ratio_mnmsup_mnmsub +  1.minority#1.supervisor#c.ln_ratio_mnmsup_mnmsub
*
*
*
* SUPERVISOR - NON-SUPERVISORY DIFFERENCE AMONG MINORITY RESPONDENT DIFFERENCES 
 
lincom  1.minority#c.ln_ratio_mnmsup_mnmsub +  1.minority#1.supervisor#c.ln_ratio_mnmsup_mnmsub - (1.minority#c.ln_ratio_mnmsup_mnmsub)



******************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************


   
*** MODEL 7: CONDITIONAL RESPONSES BY GENDER & POSITION -- GENDER BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL MODEL: [WOMEN SUPERVISORS WITHIN AGENCY j IN YEAR t / MEN SUPERVISORS WITHIN AGENCY j IN YEAR t] / [WOMEN NON-SUPERVISORS WITHIN AGENCY j IN YEAR t / MEN NON-SUPERVISORS WITHIN AGENCY j IN YEAR t] -- CONTROLLING FOR GENDER SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL ***

regress lndiversity2zeroadj  c.ln_ratio_fmsup_fmsub##i.women_het##i.supervisor    ln_ratio_fem_tot_men_tot   minority   topoffgender_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year, vce(cluster agencyid)
*
estat ic
*
** BY NON-SUPERVISORS RESPONDENT: BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN GENDERED RESPONDENTS **

lincom c.ln_ratio_fmsup_fmsub
*
*
lincom  1.women_het#c.ln_ratio_fmsup_fmsub
*
*
lincom  2.women_het#c.ln_ratio_fmsup_fmsub
*
*
*
lincom 2.women_het#c.ln_ratio_fmsup_fmsub -  1.women_het#c.ln_ratio_fmsup_fmsub
*
*
*
*


** BY SUPERVISOR RESPONDENT:BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEENGENDERED RESPONDENTS **

lincom c.ln_ratio_fmsup_fmsub + 1.supervisor#c.ln_ratio_fmsup_fmsub
*
*
lincom  1.women_het#c.ln_ratio_fmsup_fmsub +  1.women_het#1.supervisor#c.ln_ratio_fmsup_fmsub
*
*
lincom  2.women_het#c.ln_ratio_fmsup_fmsub +  2.women_het#1.supervisor#c.ln_ratio_fmsup_fmsub
*
*
*
lincom  2.women_het#c.ln_ratio_fmsup_fmsub +  2.women_het#1.supervisor#c.ln_ratio_fmsup_fmsub - (1.women_het#c.ln_ratio_fmsup_fmsub +  1.women_het#1.supervisor#c.ln_ratio_fmsup_fmsub)



* SUPERVISOR - NON-SUPERVISORY DIFFERENCE AMONG WOMEN RESPONDENT DIFFERENCES [NON-MINORITY WOMEN RESPONDENTS FOLLOWED BY MINORITY WOMEN RESPONDENTS] -- DO NOT PLOT IN GRAPHS [ONLY FOR TEXT]!

lincom  1.women_het#c.ln_ratio_fmsup_fmsub +  1.women_het#1.supervisor#c.ln_ratio_fmsup_fmsub  - (1.women_het#c.ln_ratio_fmsup_fmsub)
*
lincom  2.women_het#c.ln_ratio_fmsup_fmsub +  2.women_het#1.supervisor#c.ln_ratio_fmsup_fmsub  - (2.women_het#c.ln_ratio_fmsup_fmsub)






   
*** MODEL 8: CONDITIONAL RESPONSES BY RACE/ETHNICITY & POSITION -- RACIAL/ETHNIC BETWEEN-IDENTITY GROUP STATUS DIFFERENTIAL MODEL: [MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY SUPERVISORS WITHIN AGENCY j IN YEAR t] / [MINORITY NON-SUPERVISORS WITHIN AGENCY j IN YEAR t / NON-MINORITY NON-SUPERVISORS WITHIN AGENCY j IN YEAR t] -- CONTROLLING FOR RACIAL/ETHNIC SUPERVISORY EMPLOYEE IDENTITY GROUP DIFFERENTIAL  ***

regress  lndiversity2zeroadj  c.ln_ratio_mnmsup_mnmsub##i.minority_het##i.supervisor  ln_ratio_min_tot_nmin_tot   gender  topoffminority_2 lntotworkforce_count  ln_professionals_total_ratio  i.agencyid i.year if e(sample), vce(cluster agencyid)
*
estat ic
*

** BY NON-SUPERVISORS RESPONDENT: BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN MINORITY/NON-MINORITY RESPONDENTS **

lincom c.ln_ratio_mnmsup_mnmsub
*
*
lincom  1.minority_het#c.ln_ratio_mnmsup_mnmsub
*
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub
*
*
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub - 1.minority_het#c.ln_ratio_mnmsup_mnmsub

*
*
*
*
*


** BY SUPERVISOR RESPONDENT: BETWEEN-IDENTITY "IN" GROUP VERSUS "OUT" GROUP STATUS DIFFERENTIAL BETWEEN MINORITY/NON-MINORITY RESPONDENTS **

lincom c.ln_ratio_mnmsup_mnmsub+ 1.supervisor#c.ln_ratio_mnmsup_mnmsub
*
*
lincom  1.minority_het#c.ln_ratio_mnmsup_mnmsub +  1.minority_het#1.supervisor#c.ln_ratio_mnmsup_mnmsub
*
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub +  2.minority_het#1.supervisor#c.ln_ratio_mnmsup_mnmsub
*
*
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub +  2.minority_het#1.supervisor#c.ln_ratio_mnmsup_mnmsub - (1.minority_het#c.ln_ratio_mnmsup_mnmsub +  1.minority_het#1.supervisor#c.ln_ratio_mnmsup_mnmsub)
*
*
*
*


* SUPERVISOR - NON-SUPERVISORY DIFFERENCE AMONG MINORITY RESPONDENT DIFFERENCES [MINORITY MEN RESPONDENTS FOLLOWED BY MINORITY WOMEN RESPONDENTS] -- DO NOT PLOT IN GRAPHS [ONLY FOR TEXT]!

lincom  1.minority_het#c.ln_ratio_mnmsup_mnmsub +  1.minority_het#1.supervisor#c.ln_ratio_mnmsup_mnmsub - (1.minority_het#c.ln_ratio_mnmsup_mnmsub)
*
lincom  2.minority_het#c.ln_ratio_mnmsup_mnmsub +  2.minority_het#1.supervisor#c.ln_ratio_mnmsup_mnmsub - (2.minority_het#c.ln_ratio_mnmsup_mnmsub)


save "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\2010-2019_DATA FINAL.08-07-2024.post-estimation.dta", replace 


************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************
************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************

clear

*** FIGURE 1: PLOT ELASTICITY COEFFICIENTS FOR MODELS 1-2 USING LINCOMS ABOVE FOR EACH MODEL [4 VERTICAL PLOTS]:  ///

import excel "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\figure1.xlsx", sheet("Sheet1") firstrow
destring, replace

set scheme sj, permanently 
graph set window fontface "Century Schoolbook"

twoway (rcap low95 high95 row, vert) (scatter estimates row if group ==1, msymbol(square) mcolor(orange))(scatter estimates row if group ==2, msymbol(circle_hollow) mcolor(orange))(scatter estimates row if group ==3, msymbol(square) mcolor(navy))(scatter estimates row if group ==4, msymbol(circle_hollow) mcolor(navy)), legend(row(1) order(2 "Gender" 4 "Race/Ethnicity") pos(6)) title("FIGURE 1" "Relationship Between Authority Differentials and D&I Employee Evaluations" "(By Respondent Single Social Identity Group)", size(medsmall)) ylabel(-0.1(.1)0.3, labsize (small) angle(horizon)) xtitle("AD Estimates [Differentials by Respondent Single Social Identity Group]" "(Models 1&2)", size(small)) xlabel("", noticks) yline(0, lpattern(dash) lcolor(gs8)) aspect(.5) 

clear

*** FIGURE 2: PLOT ELASTICITY COEFFICIENTS FOR MODELS 3-4 USING LINCOMS ABOVE FOR EACH MODEL [8 VERTICAL PLOTS]:  ///

import excel "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\figure2.xlsx", sheet("Sheet1") firstrow
destring, replace

set scheme sj, permanently 
graph set window fontface "Century Schoolbook"

twoway (rcap low95 high95 row, vert) (scatter estimates row if group ==1, msymbol(square) mcolor(orange))(scatter estimates row if group ==2, msymbol(circle_hollow) mcolor(orange)) (scatter estimates row if group ==3, symbol(diamond_hollow) mcolor(orange))(scatter estimates row if group ==4, symbol(triangle_hollow) mcolor(orange))(scatter estimates row if group ==5, symbol(square) mcolor(navy))(scatter estimates row if group ==6, symbol(circle_hollow) mcolor(navy))(scatter estimates row if group ==7, symbol(diamond_hollow) mcolor(navy))(scatter estimates row if group ==8, symbol(triangle_hollow) mcolor(navy)), legend(row(1) order(2 "Gender" 6 "Race/Ethnicity") pos(6)) title("FIGURE 2"  "Relationship Between Authority Differentials and D&I Employee Evaluations" `"(By Respondent Intersectional Social Identity Group)"', size(medsmall)) ylabel(-0.1(.1)0.3, labsize (small) angle(horizon)) xtitle("Gender AD Effects: by Respondent Intersectionality Group   Race/Ethnicity AD Effects: by Respondent Intersectionality Group", size(vsmall)) xlabel("", noticks) yline(0, lpattern(dash) lcolor(gs8)) aspect(.5)

clear

*** FIGURE 3: PLOT ELASTICITY COEFFICIENTS FOR MODELS 5-8 USING LINCOMS ABOVE FOR EACH MODEL [12 VERTICAL PLOTS]: NON-SUPERVISOR RESPONDENT ESTIMATES: ///

import excel "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\figure3.xlsx", sheet("Sheet1") firstrow
destring, replace

set scheme sj, permanently  
graph set window fontface "Century Schoolbook"

twoway (rcap low95 high95 row, vert) (scatter estimates row if group ==1, msymbol(square) mcolor(orange))(scatter estimates row if group ==2, msymbol(square_hollow) mcolor(orange)) (scatter estimates row if group ==3, msymbol(square) mcolor(navy))(scatter estimates row if group ==4, msymbol(square_hollow) mcolor(navy))(scatter estimates row if group ==5, msymbol(square) mcolor(orange))(scatter estimates row if group ==6, msymbol(circle_hollow) mcolor(orange))(scatter estimates row if group ==7, msymbol(diamond_hollow) mcolor(orange))(scatter estimates row if group ==8, msymbol(triangle_hollow) mcolor(orange))(scatter estimates row if group ==9, msymbol(square) mcolor(navy))(scatter estimates row if group ==10, msymbol(circle_hollow) mcolor(navy))(scatter estimates row if group ==11, msymbol(diamond_hollow) mcolor(navy))(scatter estimates row if group ==12, msymbol(triangle_hollow) mcolor(navy)), legend(row(1) order(2 "Gender" 4 "Race/Ethnicity") pos(6)) xlabel("", noticks) title("FIGURE 3" "Relationship Between Authority Differentials and D&I Employee Evaluations" "(Non-Supervisory Respondents: Single and Intersectional Social Identity Groups)", size(medsmall)) ylabel(-0.1(.1)0.2, labsize (small) angle(horizon)) xtitle("AD Effects: by Respondent Single Identity Group     AD Effects: by Respondent Intersectionality Group", size(vsmall)) xlabel("", noticks) yline(0, lpattern(dash) lcolor(gs8)) aspect(.5)

clear

*** FIGURE 4: PLOT ELASTICITY COEFFICIENTS FOR MODELS 5-8 USING LINCOMS ABOVE FOR EACH MODEL [12 VERTICAL PLOTS]: SUPERVISOR RESPONDENT ESTIMATES: ///

import excel "C:\Users\jungy\Dropbox\DISCRIMINATION PROJECT\Organizational Diversity\Statistics\figure4.xlsx", sheet("Sheet1") firstrow
destring, replace

set scheme sj, permanently 
graph set window fontface "Century Schoolbook"

twoway (rcap low95 high95 row, vert) (scatter estimates row if group ==1, msymbol(square) mcolor(orange))(scatter estimates row if group ==2, msymbol(square_hollow) mcolor(orange)) (scatter estimates row if group ==3, msymbol(square) mcolor(navy))(scatter estimates row if group ==4, msymbol(square_hollow) mcolor(navy))(scatter estimates row if group ==5, msymbol(square) mcolor(orange))(scatter estimates row if group ==6, msymbol(circle_hollow) mcolor(orange))(scatter estimates row if group ==7, msymbol(diamond_hollow) mcolor(orange))(scatter estimates row if group ==8, msymbol(triangle_hollow) mcolor(orange))(scatter estimates row if group ==9, msymbol(square) mcolor(navy))(scatter estimates row if group ==10, msymbol(circle_hollow) mcolor(navy))(scatter estimates row if group ==11, msymbol(diamond_hollow) mcolor(navy))(scatter estimates row if group ==12, msymbol(triangle_hollow) mcolor(navy)), legend(row(1) order(2 "Gender" 4 "Race/Ethnicity") pos(6)) xlabel("", noticks) title("FIGURE 4" "Relationship Between Authority Differentials and D&I Employee Evaluations" "(Supervisor Respondents: Single and Intersectional Social Identity Groups)", size(medsmall)) ylabel(-0.1(.1)0.3, labsize (small) angle(horizon)) xtitle("AD Effects: by Respondent Single Identity Group     AD Effects: by Respondent Intersectionality Group", size(vsmall)) xlabel("", noticks) yline(0, lpattern(dash) lcolor(gs8)) aspect(.5)





************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************
************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************************


log close
